Pensieve header: A unified verification notebook for the PPSA project.

Continues pensieve://2017-06/ and pensieve://2017-08/.

Prolog

```
Go;
```

```
SetOptions[EvaluationNotebook[],
  NotebookEventActions → {
    {"MenuCommand", "Save"} :→ (
      SetDirectory["C:\\drorbn\\AcademicPensieve\\Projects\\PPSA"];
      Get["MakeVSnips.m"]
     ),
    PassEventsDown → True
   }];
```

Initialization

The “degree carrier / filtration parameter” is $\hbar$, and all “coupling constants” are proportional to it.

TD

```
$TħD = 5; $TϵD = 2; ϵ /: ϵ^d_. /; d > $TϵD := 0;
Simp[ℰ_] := Collect[ℰ, _CU | _QU, Expand];
```

DeclareAlgebra

QLImplementation

```
ħ /: ħ^d_. /; d > $TħD := 0;
Unprotect[NonCommutativeMultiply];
NonCommutativeMultiply[x_] := x; 0 ** _ = _ ** 0 = 0;
B[x_, x_] = 0; B[x_, y_] := x ** y - y ** x;
(x_Plus) ** y_ := (# ** y) & /@ x; x_ ** (y_Plus) := (x ** #) & /@ y;
```

QLImplementation

```
DeclareAlgebra[U_Symbol, opts__Rule] := Module[{gp, sr, cp, S,
    gs = Generators /. {opts}, cs = Centrals /. {opts}
  },
  gp = Alternatives @@ gs;
  gp = gp | gp_; (* generators pattern *)
  sr = Thread[gs → Range@Length@gs]; (* sorting rule *)
  cp = Alternatives @@ cs; (* centrals pattern *)
  S[ℰ_] := Collect[ℰ, _U, Expand];
  U_i_[ℰ_] := ℰ /. {t : cp ⧴ t_i, u_U ⧴ Replace[u, x_ ⧴ x_i, 1]};
  B[U@(x_)_i_, U@(y_)_i_] := B[U@x_i, U@y_i] = U_i@B[U@x, U@y];
  B[U@(x_)_i_, U@(y_)_j_] /; i =!= j := 0;
  B[U@y_, U@x_] := S[-B[U@x, U@y]];
  x_ ** U[] := x; U[] ** x_ := x;
  (a_ * x_U) ** (b_ * y_U) := If[a b === 0, 0, S[a b (x ** y)]];
  (a_ * x_U) ** y_ := S[a (x ** y)]; x_ ** (a_ * y_U) := S[a (x ** y)];
  U[xx___, x_] ** U[y_, yy___] := If[OrderedQ[{x, y} /. sr],
    U[xx, x, y, yy], U@xx ** (U@y ** U@x + B[U@x, U@y]) ** U@yy];
  U@{c_. * (l : gp)^n_, r___} /; FreeQ[c, gp] := S[c U@Table[l, {n}] ** U@{r}];
  U@{c_. * l : gp, r___} := S[c U[l] ** U@{r}];
  U@{c_, r___} /; FreeQ[c, gp] := S[c U@{r}];
  U@{} = U[];
  U@{l_Plus, r___} := S[U@{#, r} & /@ l];
  U@{l_, r___} := U@{Expand[l], r};
  𝕆_U[poly_, specs___] := Module[{sp, null, vs, us, z},
    sp = Replace[{specs}, l_List ⧴ (l → null), {1}];
    vs = Join @@ (First /@ sp);
    us = Join @@ (sp /. (l_ → s_) ⧴ (l /. x_i_ ⧴ x_s));
    S[Total[
      CoefficientRules[
        Normal@Series[poly, {ℏ, 0, $TℏD}], vs] /. (p_ → c_) ⧴ c U@(us^p)
      ]] /. x_null ⧴ x
    ]
  ]
```

Implementing $sl_2^{\gamma\epsilon}$

CU

```
DeclareAlgebra[CU, Generators → {y, a, x}, Centrals → {t}];
B[CU@a, CU@y] = -γ CU@y; B[CU@x, CU@a] = -γ CU@x;
B[CU@x, CU@y] = 2 ϵ CU@a - t CU[];
```

Verifying associativity on triples of generators:

```
With[{bas = CU /@ {y, a, x}},
 Table[z1 ** (z2 ** z3) - (z1 ** z2) ** z3 // Simp,
  {z1, bas}, {z2, bas}, {z3, bas}]]
```

```
{{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}},
 {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}, {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}}
```

Verifying associativity on a “random” triple:

```
With[{z1 = CU[y, y, a, a, x, x], z2 = CU[y, a, x], z3 = CU[y, y, a, x]}, {
   (rhs = (z1 ** z2) ** z3 // Simp) // Short,
   z1 ** (z2 ** z3) - rhs // Simp
  }] // Timing
```

$\{0.46875, \{(28\,t^2\,\gamma^4 + 116\,t\,\gamma^5\,\epsilon + 120\,\gamma^6\,\epsilon^2)\,\mathrm{CU}[y, y, y, x, x] + (4\,t^3\,\gamma + 8\,t^2\,\gamma^2\,\epsilon + 4\,t\,\gamma^3\,\epsilon^2)$
$\mathrm{CU}[y, y, a, a, a, x] + \ll 22 \gg + \mathrm{CU}[y, y, y, y, y, a, a, a, a, x, x, x, x], 0\}\}$

Implementing $\mathcal{U}_{\gamma\epsilon;\hbar}$

With $q = e^{\hbar\gamma\epsilon}$, $A = e^{-\hbar\epsilon a}$, and $[f, g]_q := fg - qgf$, our algebra is $\mathcal{U}_{\gamma\epsilon;\hbar} = \langle t, y, a, x\rangle/\mathcal{R}$, where $\mathcal{R} = ([t, *] = 0,\ [a, y] = -\gamma y,\ [x, y]_q = (1 - T^2 A^2)/\hbar,\ [x, a] = -\gamma x)$.

QU

```
DeclareAlgebra[QU, Generators → {y, a, x}, Centrals → {t, T}];
q := Normal@Series[e^(γ ϵ ħ), {ħ, 0, $TħD}];
B[QU@a, QU@y] = -γ QU@y; B[QU@x, QU@a] = -γ QU@x;
B[QU@x, QU@y] := (q - 1) QU@{y, x} + O_QU[(1 - T^2 e^(-2 ħ ϵ a)) / ħ, {a}];
```

```
With[{bas = QU /@ {y, a, x}}, Table[z1 ** z2 - z2 ** z1 // Simp, {z1, bas}, {z2, bas}]]
```

$\{\{0, \gamma\,\mathrm{QU}[y], \left(-\frac{1}{\hbar} + \frac{T^2}{\hbar}\right)\mathrm{QU}[\,] - 2\,T^2\,\epsilon\,\mathrm{QU}[a] + 2\,T^2\,\epsilon^2\,\hbar\,\mathrm{QU}[a, a] + \left(-\gamma\,\epsilon\,\hbar - \frac{1}{2}\gamma^2\,\epsilon^2\,\hbar^2\right)\mathrm{QU}[y, x]\},$
$\{-\gamma\,\mathrm{QU}[y], 0, \gamma\,\mathrm{QU}[x]\},$
$\{\left(\frac{1}{\hbar} - \frac{T^2}{\hbar}\right)\mathrm{QU}[\,] + 2\,T^2\,\epsilon\,\mathrm{QU}[a] - 2\,T^2\,\epsilon^2\,\hbar\,\mathrm{QU}[a, a] + \left(\gamma\,\epsilon\,\hbar + \frac{1}{2}\gamma^2\,\epsilon^2\,\hbar^2\right)\mathrm{QU}[y, x], -\gamma\,\mathrm{QU}[x], 0\}\}$

Verifying associativity on triples of generators:

```
With[{bas = QU /@ {y, a, x}},
 Table[z1 ** (z2 ** z3) - (z1 ** z2) ** z3 // Simp,
  {z1, bas}, {z2, bas}, {z3, bas}]]
```

```
{{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}},
 {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}, {{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}}
```

Verifying associativity on a “random” triple (~34 secs):

```
With[{z1 = QU[y, y, a, a, x, x], z2 = QU[y, a, x], z3 = QU[y, y, a, x]}, {
   (rhs = (z1 ** z2) ** z3 // Simp) // Short,
   z1 ** (z2 ** z3) - rhs // Simp
  }] // Timing
```

```
{34., {<<1>>, 0}}
```

Verifying that $\text{Lim}_{\hbar\to 0}$ QU = CU using a “random” product (~23 secs):

```
With[{z1 = CU[y, y, a, a, x, x], z2 = CU[y, a, x], z3 = CU[y, y, a, x]}, {
   Short[lhs = z1 ** (z2 ** z3)],
   Short[rhs = (QU @@ z1) ** ((QU @@ z2) ** (QU @@ z3))],
   Expand[Limit[rhs /. {QU → CU, T → e^(ħ t/2)}, ħ → 0] == lhs]
  }] // Timing
```

$\{22.5313, \{2\,(8\,t^2\,\gamma^4 + 16\,t\,\gamma^5\,\epsilon)\,\text{CU}[y,y,y,x,x] + (8\,t\,\gamma^5\,\epsilon + 16\,\gamma^6\,\epsilon^2)\,\text{CU}[y,y,y,x,x,x] + \ll 93\gg +$
$\text{CU}[y,y,y,y,y,a,a,a,a,x,x,x,x],\ (-8\,T^2\,\gamma^6\,\epsilon^2 + 8\,T^4\,\gamma^6\,\epsilon^2)\,\text{QU}[y,y,y,x,x] +$
$\ll 422\gg + \left(\gamma\,\epsilon\,\hbar + \frac{15}{2}\,\gamma^2\,\epsilon^2\,\hbar^2\right)\text{QU}[y,y,y,y,y,a,a,a,a,x,x,x,x],\ \text{True}\}\}$

Implementing θ

theta

```
θ[CU@y] = -CU@x; θ[CU@a] = -CU@a; θ[CU@x] = -CU@y;
```

theta

```
θ[QU@y] = 𝕆QU[-T^-1 e^(ħ ϵ a) x, {a, x}]; θ[QU@a] = -QU@a; θ[QU@x] = 𝕆QU[-T^-1 e^(ħ ϵ a) y, {a, y}];
```

theta

```
θ[(u : CU | QU)[]] = u[];
θ[(u : CU | QU)[v_i_]] := u_i[θ[v]];
θ[(u : CU | QU)[vs__]] := NonCommutativeMultiply @@ (θ /@ u /@ {vs});
θ[ℰ_] := Simp[ℰ /. {t → -t, T → T^-1, u_CU | u_QU :→ θ[u]}];
```

Verifying involutivity on CU:

```
With[{bas = CU /@ {y, a, x}},
 Table[{z, θ[z], θ[θ[z]]}, {z, bas}] ]
```

{{CU[y], -CU[x], CU[y]}, {CU[a], -CU[a], CU[a]}, {CU[x], -CU[y], CU[x]}}

Verifying that θ is a multiplicative homomorphism on CU:

```
With[{bas = CU /@ {y, a, x}},
 Table[θ[z1 ** z2] - θ[z1] ** θ[z2], {z1, bas}, {z2, bas}] ]
```

{{0, 0, 0}, {0, 0, 0}, {0, 0, 0}}

Verifying involutivity on QU:

```
With[{bas = QU /@ {y, a, x}},
 Table[z → θ[z] → θ[θ[z]], {z, bas}] ]
```

$\left\{\text{QU}[y] \to -\frac{\text{QU}[x]}{T} - \frac{\epsilon\,\hbar\,\text{QU}[a,x]}{T} - \frac{\epsilon^2\,\hbar^2\,\text{QU}[a,a,x]}{2\,T} \to \text{QU}[y],\ \text{QU}[a] \to -\text{QU}[a] \to \text{QU}[a],\right.$
$\left.\text{QU}[x] \to \left(-\frac{1}{T} + \frac{\gamma\,\epsilon\,\hbar}{T} - \frac{\gamma^2\,\epsilon^2\,\hbar^2}{2\,T}\right)\text{QU}[y] + \left(-\frac{\epsilon\,\hbar}{T} + \frac{\gamma\,\epsilon^2\,\hbar^2}{T}\right)\text{QU}[y,a] - \frac{\epsilon^2\,\hbar^2\,\text{QU}[y,a,a]}{2\,T} \to \text{QU}[x]\right\}$

Verifying that θ is a multiplicative homomorphism on QU:

```
With[{bas = QU /@ {y, a, x}},
 Table[{z1, z2} → Simp[θ[z1 ** z2] - θ[z1] ** θ[z2]], {z1, bas}, {z2, bas}] ]
{{{QU[y], QU[y]} → 0, {QU[y], QU[a]} → 0, {QU[y], QU[x]} → 0},
 {{QU[a], QU[y]} → 0, {QU[a], QU[a]} → 0, {QU[a], QU[x]} → 0},
 {{QU[x], QU[y]} → 0, {QU[x], QU[a]} → 0, {QU[x], QU[x]} → 0}}
```